

# Worksheets

Daniel Hillerström

6th August 2015

# Contents

<b>1</b>	<b>Introduction</b>	<b>3</b>
1.1	Problem analysis . . . . .	3
1.1.1	The problem with monads . . . . .	4
1.1.2	Composing monads with Monad Transformers . . . . .	5
1.2	Problem statement . . . . .	8
1.3	Related work . . . . .	9
1.3.1	The Eff language . . . . .	9
1.3.2	Haskell libraries . . . . .	9
1.3.3	Frank . . . . .	10
1.3.4	Idris' Effects . . . . .	10
1.3.5	Koka with row polymorphic effects . . . . .	10
1.4	Proposed solution . . . . .	11
1.4.1	Objectives, aim and scope . . . . .	11
1.4.2	Contributions . . . . .	11
<b>2</b>	<b>Background</b>	<b>12</b>
2.1	Row polymorphism . . . . .	12
<b>3</b>	<b>Programming with handlers in Links</b>	<b>14</b>
3.1	Closed handlers . . . . .	14
3.1.1	Transforming the results of computations . . . . .	15
3.1.2	Exception handling . . . . .	16
3.1.3	Interpreting Nim . . . . .	18
3.2	Open handlers . . . . .	24
3.2.1	An effectful coffee dispenser in Links . . . . .	24
3.2.2	Reinterpreting Nim . . . . .	26
	<b>Bibliography</b>	<b>27</b>

# Chapter 1

## Introduction

Some introductory text.

### 1.1 Problem analysis

Used car-dealers got a notorious reputation for being dishonest. Dishonest, because, they are reluctant to disclose any unfortunate effects that their cars may have. On the other hand, if they *did* disclose all the effects then we probably would not buy them anyway as it appears too much effort to handle those effects.

Correspondingly, some programming languages do not disclose the potential run-time effects of code execution, e.g. the ML-family of languages. For example consider the signature `readFile : string -> [string]` for a function in SML, its suggestive name hints that given a file name the function reads the file and return the contents line by line. In order to read a file the function must inevitably perform a side-effecting action, namely, accessing a storage media. But this information is not conveyed in the function signature.

Other languages disclose effects, albeit with varying degree. For example the Java programming language requires programmers to be explicit about potential unhandled checked exceptions that may occur during run-time, e.g. `String[] readFile(String f) throws IOException`. But programmers can circumvent this requirement by raising unchecked exceptions. Critics argue that Java's checked exceptions suffer versionability and scalability issues [2], and therefore it is better not to have explicit `throws` declarations.

The Haskell programming language is also explicit about effects, but, in contrast to Java, it offers no escape hatch to be implicit. Haskell insists that every effectful computation is encapsulated inside an appropriate monad. In Haskell the file reading function would be typed as `readFile :: String -> IO [String]`, where the `IO`-annotation signifies that the function might perform an input/output side-effect. A consequence of enforcing

effectful computations to be wrapped inside a monad is that the function signature conveys additional information about the computation which the programmer and compiler can rely on.

### 1.1.1 The problem with monads

Monads provide a remarkably powerful way for structuring computations, because they integrate effectful and pure computations in an elegant and flexible manner. Sadly, monads do not compose well [8], and accordingly it is difficult to give a monadic description of computations that might perform multiple effects. Consider the following attempt at modelling a coffee dispenser in Haskell:

**Example 1** (Coffee dispenser using monads). First we define the sum type `Dispensable` which has two labels: `Coffee` and `Tea`. They represent the two items that the coffee machine can dispense.

```
data Dispensable = Coffee | Tea deriving Show

type ItemCode    = Integer
type Inventory   = [(ItemCode,Dispensable)]
inventory = [(1,Coffee),(2,Tea)]
```

The `ItemCode` type models a button on the coffee machine, and `Inventory` associates buttons with dispensable items. The `inventory` is fixed, i.e. it will not change during run-time. We can make this explicit by encapsulating the `inventory` inside a `Reader`-monad, e.g.

```
dispenser :: ItemCode → Reader Inventory (Maybe
    Dispensable)
dispenser n = do inv ← ask
    let item = lookup n inv
    return item
```

The type `Reader Inventory (Maybe Dispensable)` tells us that `dispenser` accesses a read-only instance of `Inventory` and maybe returns an instance of `Dispensable`. The `Maybe`-type captures the possibility of failure, e.g. if the user requests an item that is not in the inventory. The monadic operation `ask` retrieves the inventory from the `Reader`-monad and `lookup` checks whether the item `n` is in the inventory.

Although, `Maybe` is a monad we cannot use its monadic interface, because we are in the context of the `Reader`-monad. For this simple computation it is not an issue, but it would be desirable to be able to use the failure handling capabilities of the `Maybe`-monad.  $\square$

Imagine that we want log when tea or coffee is being dispensed. The `Writer`-monad provide such capabilities. It is not immediately clear how we can integrate `Writer` with our model. Ideally, we would want a monadic computation like:

```

do inv ← ask
  item ← lookup n inv
  tell ◦ show $ item
  return item

```

Here the monadic operation `tell` writes to the medium contained in the `Writer`-monad. However, this code does not type check. A moment's thought reveals that using just monads there is no way to construct a type for that expression. The type we want is something like

$$\text{Writer } w \sqcap \text{Reader } e \sqcap \text{Maybe Dispensable}$$

where `w` is the type of the writable medium, `e` is the type of an environment and  $\sqcap$  is some “type-glue” that joins the types together. This type is exactly a Monad Transformer type which we discuss in Section 1.1.2. But using regular monads it is not possible to construct this type. Let us desugar the above expression to see why:

```

ask >>= \inv →
  lookup n inv >>= \item →
    tell ◦ show $ item
    >> return item

```

The bind operator (`>>=`) is the problem as its type is

$$\text{Monad } m \Rightarrow m \ a \rightarrow (a \rightarrow m \ b) \rightarrow m \ b$$

Essentially, this type tells us that we cannot compose monads of different types as the monad type `m` is fixed throughout the computation. Thus we see that monads lack compositionality and modularity in general.

### Effect granularity

It is possible to solve the problem using regular monads. However, it comes at a cost as suggested by the type signature of the bind operator we can compose one monad with another as long as they got the same monadic type. So, we could just use one monadic type to describe all effects. It is very tempting to bake everything into an `IO`-monad as we possibly want to I/O capabilities at some point. Albeit, `IO` is a very conservative estimate on which effects our computation might perform. Consequently, we get coarse-grained effect signatures as opposed to more specific, fine-grained effect signatures.

### 1.1.2 Composing monads with Monad Transformers

Monad Transformers allow two monads to be combined by stacking one on top of the other. Furthermore, a Monad Transformer is itself a monad, and thus we can create arbitrarily complex compositions. Incidentally, Monad

Transformers can capture computations that may cause several different effects. The following example rewrites the coffee dispenser model from Example 1 using Monad Transformers.

**Example 2** (Coffee dispenser using Monad Transformers). Most monads have a Monad Transformer cousin; by convention Monad Transformers have a capital T suffix, e.g. the `Reader`-monad’s transformer is named `ReaderT`.

We rewrite Example 1 to use the `WriterT`, and `ReaderT` monad instead of `Reader`:

```
dispenser1
:: ItemCode →
   WriterT String (ReaderT Inventory Maybe)
   Dispensable

dispenser1 n = do inv ← lift ask
                  item ← lift ∘ lift $ lookup n inv
                  tell ∘ show $ item
                  return item
```

The type may look dubious. Basically, we have built a Monad Transformer stack with three monads:

- Top of the stack: `WriterT` with a writable medium of type `String`.
- Middle: `ReaderT` with read access to an environment of type `Inventory`.
- Bottom: `Maybe` provides exception handling capabilities.

Monad Transformers allow us to express something reminiscent of the monadic computation we sought in Example 1. It is worth noting that we now use `Maybe` as a monad as opposed to an ordinary value. The benefits are obvious as we get the exception handling capabilities of `Maybe` for “free”.

However, it is not entirely free as we have to introduce `lift` operations. The `lift` operations are necessary in order to work with a specific effect down the transformer stack. For example in order to use `ask` we have to `lift` once as the `ReaderT` is the second type in the stack. Moreover, to use the monadic capabilities of `Maybe` we have to `lift` twice because it is at the bottom of the stack. Using `tell` requires no `lifts` in this example as `WriterT` is the top type. Consider what happens when we add yet another monad to the stack:

```
dispenser2
:: ItemCode →
   RandT StdGen (WriterT String (ReaderT Inventory
   Maybe)) Dispensable

dispenser2 n
= do r ← getRandomR (1,20)
```

```

inv ← lift ∘ lift $ ask
item ← lift ∘ lift ∘ lift $ lookup' r n inv
lift ∘ tell ∘ show $ item
return item
where
  lookup' r n inv = if r > 10
                    then lookup n inv
                    else Nothing

```

Here we extended our model with randomness to capture the possibility of failure caused by the system rather than the user. The `RandT` monad provides random capabilities. Moreover, we added it to the top of the transformer stack. Accordingly, we now have to use an additional `lift`, in particular, we have to `lift` in order to use `tell` now.  $\square$

Example 2 demonstrates that we can compose monads at the cost of lifting. We can think of a `lift` operation as “peeling off a layer” of the transformer stack. Thus the transformer stack enforces a static ordering on effects and interactions between effect layers [9].

Furthermore, the ordering leaks into the type signature which complicates modularity. For example, we may have a function which takes as input an effectful computation with type signature, say, `WriterT w Reader e a`. Now, the actual effectful computation has to have a type signature with the *exact* same ordering of effects even though `Writer` and `Reader` commute, i.e. the following types are isomorphic:

$$\text{WriterT } w \text{ Reader } e \text{ a} \simeq \text{ReaderT } e \text{ Writer } w \text{ a}.$$

So, we would have to permute the type signature of the actual computation [7], e.g.

```
permute :: ReaderT e Writer w a → WriterT w Reader e a
```

In this case it is safe because the two monads commute. But in general monads do not commute and therefore the consequence of permuting monads can be severe as we shall see in the next section.

### The ordering implies the semantics

The effect ordering hard wires the semantics and syntactical structure of computations. Consider the following example adapted from O’Sullivan et. al [3]:

**Example 3** (Importance of effect ordering [3]). We will demonstrate that the `Writer` and `Maybe` monads do not commute. Let `A` be the type `WriterT String Maybe` and `B` be the type `MaybeT (Writer String)`. The two types differ in their ordering of effects; type `A` has `Writer` as its outermost effect, whilst `B` has `Maybe` as its outermost effect. Now consider the following small program that performs one `tell` operation and then fails:

```

problem :: MonadWriter String m => m ()
problem = do
  tell "this is where I fail"
  fail "oops"

```

We have two possible concrete type instantiations of `m`, namely, either `A` or `B`. But as we shall see the two types enforce different semantics:

```

ghci> runWriterT (problem :: A ())
Nothing
ghci> runWriterT $ runMaybeT (problem :: B ())
(Nothing, "this is where I fail")

```

When using type `A` we lose the result from the `tell` operation. Type `B` preserves the result. Hence the two monads do not commute, and as a result the ordering of effects determine the semantics of the computation.  $\square$

We have seen that while we gain monad compositionality with Monad Transformers we do not get modularity.

## 1.2 Problem statement

In the previous section we argued that programming with *explicit* effect is desirable, but we pointed out that it is not painless to program with explicit effects. In particular, we demonstrated that the monadic approach lacks compositionality and modularity. But we could regain compositionality using Monad Transformers, however the transformer stack imposes a statical ordering on effects which impedes modularity. Compositionality and modularity are two key properties in programming which we ideally would like to retain along with explicit effects. This observation leads us to the following problem statement:

*How may we achieve a programming model with modular, composable and unordered effects?*

Plotkin and Pretnar’s handlers for algebraic effects [10] affords a very attractive model for programming with effects. The principal idea is to decouple the semantics and syntactic structure of effectful computations, i.e. an effect is a collection of abstract operations. By abstract we mean that the operation by itself has no concrete implementation. Abstract operations compose seamlessly to form the syntactical structure of the computation, whilst handlers instantiate abstract operations with a concrete interpretation. We will discuss handlers and algebraic effects in greater detail in Section ??.

We suppose that handlers for algebraic effects provide a desirable model for programming with effects. A substantial amount of work has already been put into handlers and effects. In Section 1.3 we discuss and evaluate related work before we propose our own solution in Section 1.4.



## 1.3 Related work

This section discusses and evaluates related work on programming models with handlers and effects.

### 1.3.1 The Eff language

The *Eff* programming language, by Bauer and Pretnar [14], has a first-class implementation of handlers for algebraic effects. The language has the look and feel of OCaml. Eff achieves unordered effects through a combination of effect polymorphism and subtyping.

### 1.3.2 Haskell libraries

We will discuss two implementations of handlers on top of Haskell by Kammar et. al [8] and Wu et. al [13].

#### Data types á la carte

Swierstra [4] demonstrates how to compose effectful programs using *free monads* in Haskell. The free monads form the basis for a framework for encoding handlers and effects which the subsequent libraries use.

#### Extensible effects

A few words on Kiselyov’s paper? [9].

#### Handlers in action

Kammar et. al considers two different approaches to implement handlers on top of Haskell. One approach is based on free monads [4], whilst the other is a continuation-based approach [8].

Their handlers are encoded as type classes, thus handlers inherit the limitations of type classes. Particularly, type classes are not first-class in Haskell, so neither are the handlers. To achieve unordered effects they use type class constraints. Type classes can only be defined in top-level as Haskell do not permit local type-class definition. Consequently, every effect handler must be defined in the top-level too.

Furthermore, the order in which handlers are composed leak into the type signature, because their (open) handlers explicitly mention a parent handler [8]. Albeit, it does not cause an issue like Monad Transformer ordering issue, but it is still undesirable.

Kammar et. al hypothesises that an implementation based on row polymorphism may remedy the limitations and yield a cleaner design [8].

## Handlers in scope

Wu et. al investigate how to use handlers to delimit the scope of effects [13] as using handlers for scoping has limitations. In other words, the ordering of handlers may affect the semantics.

They present two solutions embedded in Haskell [13]. The first solution extends the existing effect handler framework based on free monads with so-called *scope markers* which fits nicely into the framework. However they demonstrate that handlers along with scope markers are insufficient to capture higher-order scoped constructs properly.

Their second approach is continuation-based and provides a *higher-order syntax* that allows to embed programs with scoping constructs [13]. However it remains an open question whether their implementation is viable in other languages than Haskell.

### 1.3.3 Frank

The Frank programming language by McBride [12] takes the notion of effect handlers to the extreme. In Frank there are no functions, there are only handlers. Moreover, it employs an interesting evaluation order known as “call-by-push-value” (CBPV). Intuitively, CBPV is the unification of the strict call-by-value and non-strict call-by-name semantics. The choice whether to employ the strict or non-strict semantics has been made explicit to the programmer.

Frank distinguishes between computations and values as a consequence side-effects can only occur in computations. Hence there is a clear separation between segments of code where effects might occur and where effects are guaranteed never to occur.

### 1.3.4 Idris’ Effects

Brady [7] presents the library EFFECTS for the dependently-typed, functional programming language IDRIS.

### 1.3.5 Koka with row polymorphic effects

Leijen’s programming language Koka is an effect-based web-oriented language [11]. It supports arbitrary user-defined effects [15]. Notably, Koka uses row polymorphism to capture unordered effects however Koka’s row polymorphism allow duplicate effect occurrences which stands in contrast to the approach we propose. In particular, Koka has no notion of effect handler except for exception handlers which are to some extent reminiscent of those in Java, C#, etc.

## 1.4 Proposed solution

Kammar et. al proposed that a row-based effect type system would yield a cleaner design [8].

### 1.4.1 Objectives, aim and scope

The aim is to examine the programming model achieved by using handlers with row polymorphic effects. In order to examine the model we must first implement it, thus the primary objective is to implement handlers and support for user-defined effects in Links.

Links has support for numerous web-oriented features, however we restrict the scope to a working implementation in top-level Links.

### 1.4.2 Contributions

The main contributions are:

- An implementation of effect handlers in Links.
- Support for row polymorphic user-defined effects in Links.
- An examination of programming with handlers and row polymorphic effects.

## Chapter 2

# Background

### 2.1 Row polymorphism

Row polymorphism is a typing discipline for records [1]. A record is an unordered collection of fields, e.g.  $\langle l_1 : t_1, \dots, l_n : t_n \rangle$  denotes a record type with  $n$  fields where  $l_i$  and  $t_i$  denote the name and type, respectively, of the  $i$ th field. Moreover, the record type is monomorphic, that is, the type is fixed. Row polymorphism, as the name suggests, makes record types polymorphic.

The following illustrates the power of row polymorphism.

**Example 4.** OCaml's (regular) record types are monomorphic.<sup>1</sup> Consider the following two record type definitions in OCaml

```
type student      = {name : string; id : string}
type supervisor = {name : string; group : string}
```

Now we can create instances of `student` and `supervisor`

```
ocaml> let daniel = {name="Daniel"; id="s1467124"};;
val daniel : person = {name="Daniel"; id="s1467124"}

ocaml> let sam = {name="Sam"; group="LFCS"};;
val sam : supervisor = {name="Sam"; group="LFCS"}
```

As expected the OCaml compiler infers the correct record types for both instances. Since both record types have the field *name* in common we might expect to define a function which prints the name field of either record type, e.g.

```
ocaml> let print_name r = r.name;;
ocaml> let () =
    print_name daniel;
    print_name sam;;
```

---

<sup>1</sup>OCaml's object types are row polymorphic.

Surprisingly this yields the following type error

```
print_name daniel;  
~~~~~  
Error: This expression has type student  
      but an expression was expected of type  
      supervisor
```

The record `daniel` is not compatible with the type of `print_name`. Because the record types are monomorphic, the compiler has to decide on compile time which record type `print_name` accepts as input parameter. Apparently in this example the compiler has decided to type `print_name` as `supervisor → string`.

Now consider the same example in Links. In contrast to OCaml record types are polymorphic in Links. First we define the `print_name` function

```
links> fun print_name(r) { r.name };  
print_name = fun : ((name:a|ρ)) ↦ a
```

Links tells us that the function accepts a record type which has *at least* the field `name`. The field type is polymorphic as signified by the presence of the type variable `a`. The additional type variable  $\rho$  is a polymorphic row variable which can be instantiated to additional fields, hence the actual input record may contain *more* fields. Now our printing function works as expected:

```
links> fun() { print_name(daniel);  
              print_name(sam) }();  
  
Daniel  
Sam
```

Here we wrapped the applications of `print_name` inside a parameterless function because Links does not support expression sequencing in the top-level.  $\square$

More on  $\rho$ , unification, etc...

## Chapter 3

# Programming with handlers in Links

### 3.1 Closed handlers

A closed handler handles a fixed set of effects, that is, it effectively describes an upper bound on which kind of effects a computation may perform. In Links this bound is made explicit in the handler's type, e.g. the closed handler `h`

```
var h = handler(m) {  
  case Op(p, _) → p  
  case Return(x) → x  
}
```

has the type  $(() \xrightarrow{\{Op:a \rightarrow a\}} a) \rightarrow a$  where the absence of a row variable in the effect signature implies that the computation `m` may not perform any other effects than `Op`. It is considered a type error to handle a computation whose effect signature is larger than the handler supports.

This restriction introduces slack into the type system. To illustrate the slack consider the following computation

```
fun comp() {  
  do Op(true);  
  if (false == true) {  
    do Op2(false)  
  } else {  
    true  
  }  
}
```

The computation `comp` has type  $() \xrightarrow{\{Op:Bool \rightarrow (), Op2:Bool \rightarrow Bool \mid \rho\}} Bool$ . Obviously, `Op2` never gets discharged. However, attempting to handle `comp` with the handler `h` yields a type error because `Op2` is present in the effect

signature of `comp`. The type system is conservative, but in general it is undecidable whether the first or second branch of a conditional expression will be taken [6].

The following sections will show increasingly interesting examples of programming with closed handlers in Links.

### 3.1.1 Transforming the results of computations

Handlers take computations as input. From a handler’s perspective a computation is a *thunk*, i.e. a parameterless function whose type is similar to  $() \xrightarrow{\{Op_i:a_i \rightarrow b_i\}} c$ . The first few examples show how to transform the output of a computation using handlers. We begin with a handler that appears to be rather boring, but in fact proves very useful as we shall see later in Section ??.

**Example 5** (The force handler). We dub the handler `force` as it takes a computation (thunk) as input, evaluates it and returns its result. It has type `force : (() → a) → a` and it is defined as

```
var force = handler(m) {
  case Return(x) → x
}
```

Essentially, this handler applies the identity transformation to the result of the computation `m`. Running `force` on a few examples should yield no surprises:

```
fun fortytwo() { 42 }
links> force(fortytwo);
42 : Int

fun hello() { "Hello" }
links> force(hello);
"Hello" : String
```

The handler `force` behaves as expected for these trivial examples. But suppose we want to print “*Hello World*” to the standard output, e.g.

```
fun print_hello() { print("Hello World") }
links> force(print_hello);
Type error: ... # Omitted for brevity
```

then the Links compiler contemptuously halts with a type error! The type of `print_hello` is  $() \rightsquigarrow ()$  which at first glance may appear to be compatible with the type of formal parameter `m`. But printing to standard output is effectful action, as indicated by the squiggly arrow in the signature, hence `print_hello` is an effectful computation. Since `force` does not handle any effects we get the type error.  $\square$

As Example 5 demonstrated the handler `force` could not handle the `print` effect caused by `print_hello`. In fact no handler in Links is able to handle `print_hello` because the `print` effect is a syntactic, built-in effect known as *wild*. Handlers only handles user-defined effects.

The next example demonstrates an actual transformation.

**Example 6** (The `listify` handler). The `listify` handler transforms the result of a handled computation into a singleton list. Its type is `listify : (() → a) → [a]` and its definition is straightforward

```
var listify = handler(m) {
  case Return(x) → [x]
}
```

When handling the computations from Example 5 we see that it behaves as expected, e.g.

```
links> listify(fortytwo);
[42] : [Int]

links> listify(hello);
["Hello"] : [String]

fun list123() { [1,2,3] }
links> listify(list123);
[[1,2,3]] : [[Int]]
```

These examples also illustrate the `Return`-case serves a similar purpose to the monadic `return`-function in Haskell whose type is `return : a → m a` for a monad *m*. It “lifts” the result into an adequate type.  $\square$

In a similar fashion to the handler `listify` in Example 6 we can define handlers that increment results by 1, perform a complex calculation using the result of the computation or wholly ignore the result. The bottom line is that it must ensure its output has an adequate type. In the case for `listify` the type must be a list of whatever type the computation yielded.

### 3.1.2 Exception handling

Until now we have only seen some simple transformations. Let us spice things up a bit. Example 7 introduces the practical handler `maybe`. It is similar to the `Maybe`-monad in Haskell. For reference we briefly sketched the behaviour of the `Maybe`-monad in Section 1.1.1.

**Example 7** (The `maybe` handler). The `maybe` handler handles one operation `Fail : a → a` that can be used to indicate that something unexpected has happened in a computation. The handler returns `Nothing` when `Fail`



is raised, and **Just** the result when the computation succeeds, thus its type is

$$\text{maybe} : (() \xrightarrow{\{\text{Fail}:a \rightarrow a\}} b) \rightarrow [| \text{Just} : b | \text{Nothing} | \rho |].$$

It is defined as

```
var maybe = handler(m) {
  case Fail(_,_) → Nothing
  case Return(x) → Just(x)
}
```

When a computation raises **Fail** the handler discards the remainder of the computation and returns **Nothing** immediately, e.g.

```
fun yikes() {
  var x = "Yikes!";
  do Fail();
  x
}
links> maybe(yikes);
Nothing() : [| Just:String | Nothing | ρ |]
```

and if the computation succeeds it transforms the result, e.g.

```
fun success() {
  true
}
links> maybe(success);
Just(true) : [| Just:Bool | Nothing | ρ |]
```

□

The next example demonstrates an alternative “exception handling strategy”.

**Example 8** (The recover handler). We can define a handler **recover** which ignores the raised exception and resumes execution of the computation. The type of **recover** is

$$\text{recover} : (() \xrightarrow{\{\text{Fail}:a \rightarrow ()\}} b) \rightarrow [| \text{Just} : b | \rho |].$$

A slight reminder here: The label **Nothing** is absent from the handler’s output type because **Just** is a polymorphic variant label and its relation to **Nothing** is conventional. We define **recover** as

```
var recover = handler(m) {
  case Fail(_,k) → k(())
  case Return(x) → Just(x)
}
```

In contrast to **maybe** from Example 7 the **recover** handler invokes the continuation **k** once. This invocation effectively resumes execution of the computation. Consider **recover** applied to the computation **yikes** from before

```
links> recover(yikes)
Just("Yikes!") : [|Just:String| $\rho$ |]
```

□

Although it is seldom a sound strategy to ignore exceptions the two Examples 7 and 8 demonstrate that we can change the semantics of the computation by changing the handler.

### 3.1.3 Interpreting Nim

Nim is a well-studied mathematical strategic game, and probably among the oldest of its kind [5]. In Nim two players take turns to pick between one and three sticks from heaps of sticks. Whoever takes the last stick wins. This play style is also known as *normal play*.

Nim enjoys many interesting game theoretic properties, however we will use a simplified version of Nim to demonstrate how handlers can give different interpretations of the same game. In our simplified version there is only one heap of  $n$  sticks. Moreover, there are two players: Alice and Bob, and Alice always starts. Our model is adapted from Kammar et. al [8].

We model the game as two mutual recursive abstract computations, e.g.

```
# Input n is the number of remaining sticks
fun aliceTurn(n) {
  if (n == 0) {
    Bob
  } else {
    var take = do Move((Alice,n));
    var r = n - take;
    bobTurn(r)
  }
}

# Symmetrically for Bob
fun bobTurn(n) {
  if (n == 0) {
    Alice
  } else {
    var take = do Move((Bob,n));
    var r = n - take;
    aliceTurn(r)
  }
}
```

The two computations are symmetrical. The input parameter `n` is the number of sticks left in the heap. First, Alice tests whether there are any sticks left, if there is not then she declares `Bob` the winner, otherwise she performs her move and then she gives the turn to Bob. The game has one abstract

operation `Move` which has the inferred type  $\text{Move} : ([\text{Alice}|\text{Bob}|\rho], \text{Int}) \rightarrow \text{Int}$ , i.e. it takes two arguments

1. Who's turn it is,
2. and the number of remaining sticks.

The operation `Move` returns the number of sticks that the current player takes. Figure 3.1 depicts the shape of the computation tree representation of the game.

Tree

Figure 3.1: Nim game computation tree

The depth of the tree is potentially infinite as the depth depends entirely on the interpretation of the operation `Move`. The inferred type of a game is

$$\text{aliceTurn} : \text{Int} \xrightarrow{\text{Move} : ([\text{Alice}|\text{Bob}|\rho], \text{Int}) \rightarrow \text{Int}} [\text{Alice}|\text{Bob}|\rho]$$

Because it is an unary function it cannot be used as an input to any handler. We rectify the problem by using a closure, i.e. we wrap the game function inside a nullary function like `fun() {aliceTurn(n)}` where `n` is a free variable captured by the surrounding context. For conveniency, we define an auxiliary function `play` to abstract away these details. It takes as input a game handler `gh` and the number of sticks at the beginning of game `n`. Moreover, the function `play` enforces the rule that Alice always starts, e.g.

```
fun play(gh, n) {
  gh(fun() {
    aliceTurn(n)
  })
}
```

The following examples demonstrates how handlers encode the strategic behaviour of the players.

**Example 9** (A naïve strategy). A very naïve strategy is to pick just *one* stick at every turn. Its implementation is straightforward

```
var naive = handler(m) {
  case Move(_, k) → k(1)
  case Return(x) → x
};
```

Here `Move` is handled uniformly. Independent of the parameterisation it always invokes the continuation `k` with 1 which corresponds to the player taking just 1 stick from the heap.

A moment's thought will tell us that we can easily predict the winner when using the `naive` strategy. The parity of  $n$ , the number of sticks at the beginning, determines the winner. For odd  $n$  Alice wins and vice versa for even  $n$ , e.g.

```
links> play(naive, 5);
Alice() : [|Alice|Bob|ρ|]

links> play(naive, 10);
Bob() : [|Alice|Bob|ρ|]

links> play(naive, 101);
Alice() : [|Alice|Bob|ρ|]
```

□

**Example 10** (Perfect vs perfect strategy). A perfect strategy makes an optimal move at each turn. An optimal move depends on the remaining number of sticks  $n$ . The perfect move can be defined as a function of  $n$ , e.g.

$$\text{perfect}(n) = \max\{n \bmod 4, 1\}$$

In our restricted Nim game a perfect strategy is a winning strategy for Alice if and only if the number of remaining sticks is *not* divisible by 4.

We implement the function `perfect` above with an addition: We pass it a continuation as second parameter

```
fun perfect(n, k) {
  k(max(mod(n,4),1))
}
```

The continuation is invoked with the optimal move. Now we can easily give a handler that assigns perfect strategies to both Alice and Bob, e.g.

```
var pvp = handler(m) {
  case Move((_,n),k) → perfect(n, k)
  case Return(x)      → x
};
```

By running some examples we see that Alice wins whenever  $n$  is not divisible by four:

```
links> play(pvp, 9);
Alice() : [|Alice|Bob|ρ|]

links> play(pvp, 18);
Alice() : [|Alice|Bob|ρ|]

links> play(pvp, 36);
Bob() : [|Alice|Bob|ρ|]
```

□

**Example 11** (Mixing strategies). A strategy often encountered in game theory is *mixing* which implies a player randomises its strategies in order to confuse its opponent. In similar fashion to **perfect** from Example 10 we define a function **mix** which chooses a strategy

```
fun mix(n,k) {
  var r = mod(nextInt(), 3) + 1;
  if (r > 1 && n ≥ r) {
    k(r)
  } else {
    k(1)
  }
}
```

The function **nextInt** returns the next integer in some random sequence. The random integer is projected into the cyclic group  $\mathbb{Z}_3 = \{0, 1, 2\}$  generated by 3. We add one to map it onto the set of valid moves  $\{1, 2, 3\}$ . If the random choice **r** is greater than the number of remaining sticks **n** then we default to take one (even though the optimal choice might be to take two).

The **mixing** strategy handler is similar to **perfect-vs-perfect** handler from Example 10

```
var mixing = handler(m) {
  case Move(_,n),k) → mix(n,k)
  case Return(x)    → x
};
```

Replaying the same game a few times ought eventually yield the two possible outcomes

```
links> play(mixing, 7);
Bob() : [|Alice|Bob|ρ|]

links> play(mixing, 7);
Alice() : [|Alice|Bob|ρ|]
```

□

**Example 12** (Brute force strategy). Examples 9-11 only invoked the continuation once per move. However, we can invoke the continuation multiple times to enumerate all possible future moves, this way we can brute force a winning strategy, if one exists. In order to brute force a winning strategy, we define a convenient utility function which computes the set of valid moves given the number of remaining sticks

```
fun validMoves(n) {
  filter(fun(m) { m ≤ n }, [1,2,3])
}
```

The function simply filters out impossible moves based on the current configuration  $n$ . Note when  $n > 3$  the function `validMoves` behaves like the identity function. The function `bruteForce` computes the winning strategy for a particular player if such a strategy exists:

```
fun bruteForce(player, n, k) {
  var winners = map(k, validMoves(n));
  var hasPlayerWon = indexOf(player, winners);
  switch (hasPlayerWon) {
    case Nothing → k(1)
    case Just(i) → k(i+1)
  }
}
```

The first line inside `bruteForce` is the critical point. Here we map the continuation `k` over the possible valid moves in the current game configuration. Thus the function effectively simulates all possible future configurations yielding a list of possible winners. The auxiliary function `indexOf` looks up the position of `player` in the list of winners. The position plus one corresponds to the winning strategy because lists indexes are zero-based. If the player has a winning strategy then the (zero-based) position is returned inside a `Just`, otherwise `Nothing` is returned.

Let Alice play the brute force strategy and let Bob play the perfect strategy which is captured by the strategy handler `bfvp`

```
var bfvp = handler(m) {
  case Move((Alice,n),k) → bruteForce(Alice,n,k)
  case Move((Bob,n),k)   → perfect(n,k)
  case Return(x)         → x
};
```

Here we use deep pattern-matching to distinguish between when `Alice` and `Bob`'s moves. Obviously, the brute force strategy is inefficient as it redoes a lot of work for each move. The winning strategy that it discovers is exactly same as the perfect strategy. Albeit, `bruteForce` computes it in exponential time whilst `perfect` computes it in constant time. The following outcomes witness that `bruteForce` and `perfect` behaves identically

```
links> play(bfvp, 9);
Alice() : [|Alice|Bob|ρ|]

links> play(bfvp, 18);
Alice() : [|Alice|Bob|ρ|]

links> play(bfvp, 36);
Bob()   : [|Alice|Bob|ρ|]
```

□

Although, the `bruteForce` strategy is significantly slower than `perfect` strategy in Example 12 the point of interest here is not efficiency but rather modularity. As Example 12 nicely demonstrates we can swap two observably equivalent handlers effortlessly. Clearly, this has practical applications for example one would be able to quickly create a prototypical system with slow but easy to implement components. Then later as the system scales one can change the slow components for a faster ones effortlessly.

Examples 9-12 gave different interpretations of the same game. Furthermore, they all computed the same thing, namely, the winner. We can use handlers to create data from computations. For example we can construct the game tree in a Nim game as Example 13 shows.

**Example 13** (Game tree generator). A node in a game tree represents a particular player's turn and an edge corresponds to a particular move. A path down the tree corresponds to a particular sequence of moves taken by the players ending in a leaf node which corresponds to the winner. Figure 3.2 shows an example game tree when starting with 3 sticks.

Our game tree is a ternary tree which we represent using a recursive variant type, e.g.

```
MoveTree = [[Take : (Player, [(Int, MoveTree)])] | Winner : Player]
```

Actually, we will not use the type explicitly, but rather rely on Links' type inference. It will infer a polymorphic variant type rather than the monomorphic type given above.

We define a function `reifyMove` which takes a player, the number of sticks, and a continuation to construct a node in the game tree, e.g.

```
fun reifyMove(player, n, k) {
  var moves = map(k, validMoves(n));
  var interval = range(1, length(moves));
  Take(player, zip(interval, moves))
}
```

First, we map the continuation over the possible valid moves in the current game configuration to enumerate the subsequent game trees. The `interval` is a list of integers from one to the number of immediate subtrees. Finally, we construct a node `Take` with `player` and the possible subsequent game trees.

We determine the winner of a particular play in the `Return`-case of the handler, e.g.

```
var mtGen = handler(m) {
  case Move((player, n), k) → reifyMove(player, n, k)
  case Return(x)           → Winner(x)
};
```

The inferred type for `mtGen` witnesses that the handler indeed constructs a tree structure:

$$\text{mtGen} : ((\ ) \xrightarrow{\{\text{Move}:(a,\text{Int})\rightarrow\text{Int}\}} b) \rightarrow \mu c. [|\text{Take} : (a, [(\text{Int}, c)])| \text{Winner} : b | \rho]$$

Figure 3.2 depicts the game tree generated by the handler when starting with 3 sticks.




Figure 3.2: Pretty print of the game tree generated by `play(mtGen, 3)`.

□

## 3.2 Open handlers

### 3.2.1 An effectful coffee dispenser in Links

In Section ?? we implemented a basic coffee dispenser model in Haskell using monads (Example 1). However, it was difficult to extend the model to include more properties like writing to a display and system failures due to monads' lack of compositionality.

In contrast (open) handlers enable us implement a modular coffee dispenser model in Links. Example 14 implements the model and Example ?? demonstrates how the modularity and compositionality of handlers lets us change the behaviour of the coffee dispenser without breaking a sweat.

**Example 14** (Coffee dispenser). The coffee dispenser performs two operations directly

1. **Ask**: Retrieves the inventory.
2. **Tell**: Writes a description of the item to some medium.



Indirectly, the coffee dispenser may perform the **Fail** operation when it looks up an item. Thus the type of the dispenser is

$$\text{dispenser} : a \xrightarrow{\{\text{Ask}:(\rightarrow[(a,b)],\text{Fail}:(\rightarrow b,\text{Tell}:b\rightarrow c|\rho)\}} c$$

We compose the coffee dispenser from the aforementioned operations and the look-up function, e.g.

```
fun dispenser(n) {
  var inv = do Ask();
  var item = lookup(n, inv);
  do Tell(item)
}
```

The monadic coffee dispenser model used three monads: **Reader**, **Writer** and **Maybe** to model the desired behaviour. We will implement three handlers which resemble the monads. First, let us implement **Reader**-monad as the handler **reader** whose type is

$$((\ ) \xrightarrow{\{\text{Ask}:(a\rightarrow[(\text{Int},[\text{Coffee}|\text{Tea}|\rho_1])])|\rho_2\}} b) - > (\ ) \xrightarrow{\{\text{Ask}:\alpha|\rho_2\}} b$$

For simplicity we hard-code the inventory into the handler

```
open handler reader(m) {
  case Ask(_,k) → k([(1,Coffee),(2,Tea)])
  case Return(x) → x
}
```

When handling the operation **Ask** the handler simply invokes the continuation **k** with the inventory as parameter. Like in Example 1 we model the inventory as an association list.

Second, we implement the handler **writer** which provide capabilities to write to a medium. We let the medium be a regular string. The handler's type is

$$((\ ) \xrightarrow{\{\text{Tell}:([\text{Coffee}|\text{Tea}]\rightarrow\text{String}|\rho)\}} a) - > (\ ) \xrightarrow{\{\text{Tell}:\alpha|\rho_2\}} a$$

and its definition is

```
open handler writer(m) {
  case Tell(Coffee,k) → k("Coffee")
  case Tell(Tea,k) → k("Tea")
  case Return(x) → x
}
```

Here we use pattern-matching to convert **Coffee** and **Tea** into their respective string representations.

Finally, we implement the **lookup** function which given a key and an association list returns the element associated with the key if the key exists in the list, otherwise it discharges the **Fail**-operation to signal failure

```

fun lookup(n, xs) {
  switch (xs) {
    case [] → do Fail()
    case ((i, e) :: xs) → if (n == i) { e }
                           else { lookup(n, xs) }
  }
}

```

To handle failure we reuse the `maybe`-handler from Section 3.1.2 with the slight change that we make it an open handler. Now, we just have to glue all the components together

```

fun runDispenser(n) {
  force(maybe(writer(reader(fun() { dispenser(n) }))))
}

```

Note, that in this example the order in which we compose handlers is irrelevant. Running a few examples we see that it behaves similarly to the monadic version we implemented in Section ??

```

links> runDispenser(1)
Just("Coffee") : [|Just:String|Nothing|ρ|]

links> runDispenser(2)
Just("Tea") : [|Just:String|Nothing|ρ|]

links> runDispenser(3)
Nothing() : [|Just:String|Nothing|ρ|]

```

□

### 3.2.2 Reinterpreting Nim

# Bibliography

- [1] Didier Rémy. “Type Inference for Records in a Natural Extension of ML”. In: *Theoretical Aspects Of Object-Oriented Programming. Types, Semantics and Language Design*. Ed. by Carl A. Gunter and John C. Mitchell. MIT Press, 1993.
- [2] Bill Venners and Bruce Eckel. *The Trouble with Checked Exceptions: A Conversation with Anders Hejlsberg, Part II*. <http://www.artima.com/intv/handcuffs.html>. 2003.
- [3] Bryan O’Sullivan, John Goerzen and Don Stewart. *Real World Haskell*. 1st. O’Reilly Media, Inc., 2008. ISBN: 0596514980, 9780596514983.
- [4] Wouter Swierstra. “Data types à la carte”. In: *Journal of Functional Programming* 18 (04 July 2008), pp. 423–436. ISSN: 1469-7653. DOI: 10.1017/S0956796808006758. URL: [http://journals.cambridge.org/article\\_S0956796808006758](http://journals.cambridge.org/article_S0956796808006758).
- [5] Anker Helms Jørgensen. “Context and Driving Forces in the Development of the Early Computer Game Nimbi”. In: *IEEE Annals of the History of Computing* 31.3 (2009), pp. 44–53. ISSN: 1058-6180.
- [6] Hans Hüttel. *Transitions and Trees: An Introduction to Structural Operational Semantics*. 1st. New York, NY, USA: Cambridge University Press, 2010. ISBN: 0521147093, 9780521147095.
- [7] Edwin Brady. “Programming and reasoning with algebraic effects and dependent types”. In: *ACM SIGPLAN International Conference on Functional Programming, ICFP’13, Boston, MA, USA - September 25 - 27, 2013*. 2013, pp. 133–144. DOI: 10.1145/2500365.2500581. URL: <http://doi.acm.org/10.1145/2500365.2500581>.
- [8] Ohad Kammar, Sam Lindley and Nicolas Oury. “Handlers in Action”. In: *Proceedings of the 18th ACM SIGPLAN International Conference on Functional Programming, ICFP ’13, Boston, Massachusetts, USA: ACM, 2013*, pp. 145–158. ISBN: 978-1-4503-2326-0. DOI: 10.1145/2500365.2500590. URL: <http://doi.acm.org/10.1145/2500365.2500590>.

- [9] Oleg Kiselyov, Amr Sabry and Cameron Swords. “Extensible Effects: An Alternative to Monad Transformers.” In: *Proceedings of the 2013 ACM SIGPLAN Symposium on Haskell*. Haskell ’13. Boston, Massachusetts, USA: ACM, 2013, pp. 59–70. ISBN: 978-1-4503-2383-3. DOI: 10.1145/2503778.2503791. URL: <http://doi.acm.org/10.1145/2503778.2503791>.
- [10] Gordon D. Plotkin and Matija Pretnar. “Handling Algebraic Effects”. In: *Logical Methods in Computer Science* 9.4 (2013). DOI: 10.2168/LMCS-9(4:23)2013. URL: [http://dx.doi.org/10.2168/LMCS-9\(4:23\)2013](http://dx.doi.org/10.2168/LMCS-9(4:23)2013).
- [11] Daan Leijen. “Koka: Programming with Row Polymorphic Effect Types”. In: *Mathematically Structured Functional Programming 2014*. EPTCS, 2014. URL: <http://research.microsoft.com/apps/pubs/default.aspx?id=210640>.
- [12] Sam Lindley and Conor McBride. *Do Be Do Be Do*. <http://homepages.inf.ed.ac.uk/slindley/papers/frankly-draft-march2014.pdf>. Draft, March 2014. 2014.
- [13] Nicolas Wu, Tom Schrijvers and Ralf Hinze. “Effect Handlers in Scope”. In: *Proceedings of the 2014 ACM SIGPLAN Symposium on Haskell*. Haskell ’14. Gothenburg, Sweden: ACM, 2014, pp. 1–12. ISBN: 978-1-4503-3041-1. DOI: 10.1145/2633357.2633358. URL: <http://doi.acm.org/10.1145/2633357.2633358>.
- [14] Andrej Bauer and Matija Pretnar. “Programming with algebraic effects and handlers”. In: *Journal of Logic and Algebraic Methods in Programming* 84.1 (2015), pp. 108–123. DOI: 10.1016/j.jlamp.2014.02.001. URL: <http://dx.doi.org/10.1016/j.jlamp.2014.02.001>.
- [15] Niki Vazou and Daan Leijen. *Remarrying effects and monads*. Submitted to ICFP ’15. <http://goto.ucsd.edu/~nvazou/koka/icfp15.pdf>. 2015.